

# NAG Library Routine Document

## G07BFF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G07BFF estimates parameter values for the generalized Pareto distribution by using either moments or maximum likelihood.

### 2 Specification

```
SUBROUTINE G07BFF (N, Y, OPTOPT, XI, BETA, ASVC, OBSVC, LL, IFAIL)
INTEGER          N, OPTOPT, IFAIL
REAL (KIND=nag_wp) Y(N), XI, BETA, ASVC(4), OBSVC(4), LL
```

### 3 Description

Let the distribution function of a set of  $n$  observations

$$y_i, \quad i = 1, 2, \dots, n$$

be given by the generalized Pareto distribution:

$$F(y) = \begin{cases} 1 - \left(1 + \frac{\xi y}{\beta}\right)^{-1/\xi}, & \xi \neq 0 \\ 1 - e^{-y/\beta}, & \xi = 0; \end{cases}$$

where

$$\beta > 0 \text{ and}$$

$$y \geq 0, \text{ when } \xi \geq 0;$$

$$0 \leq y \leq -\frac{\beta}{\xi}, \text{ when } \xi < 0.$$

Estimates  $\hat{\xi}$  and  $\hat{\beta}$  of the parameters  $\xi$  and  $\beta$  are calculated by using one of:

method of moments (MOM);

probability-weighted moments (PWM);

maximum likelihood estimates (MLE) that seek to maximize the log-likelihood:

$$L = -n \ln \hat{\beta} - \left(1 + \frac{1}{\hat{\xi}}\right) \sum_{i=1}^n \ln \left(1 + \frac{\hat{\xi} y_i}{\hat{\beta}}\right).$$

The variances and covariance of the asymptotic Normal distribution of parameter estimates  $\hat{\xi}$  and  $\hat{\beta}$  are returned if  $\hat{\xi}$  satisfies:

$$\hat{\xi} < \frac{1}{4} \text{ for the MOM;}$$

$$\hat{\xi} < \frac{1}{2} \text{ for the PWM method;}$$

$$\hat{\xi} < -\frac{1}{2} \text{ for the MLE method.}$$

If the MLE option is exercised, the observed variances and covariance of  $\hat{\xi}$  and  $\hat{\beta}$  is returned, given by the negative inverse Hessian of  $L$ .

## 4 References

Hosking J R M and Wallis J R (1987) Parameter and quantile estimation for the generalized Pareto distribution *Technometrics* **29**(3)

McNeil A J, Frey R and Embrechts P (2005) *Quantitative Risk Management* Princeton University Press

## 5 Parameters

- 1: N – INTEGER *Input*  
*On entry:* the number of observations.  
*Constraint:*  $N > 1$ .
- 2: Y(N) – REAL (KIND=nag\_wp) array *Input*  
*On entry:* the  $n$  observations  $y_i$ , for  $i = 1, 2, \dots, n$ , assumed to follow a generalized Pareto distribution.  
*Constraints:*
- $$Y(i) \geq 0.0;$$
- $$\sum_{i=1}^n Y(i) > 0.0.$$
- 3: OPTOPT – INTEGER *Input*  
*On entry:* determines the method of estimation, set:  
 OPTOPT = -2  
     For the method of probability-weighted moments.  
 OPTOPT = -1  
     For the method of moments.  
 OPTOPT = 1  
     For maximum likelihood with starting values given by the method of moments estimates.  
 OPTOPT = 2  
     For maximum likelihood with starting values given by the method of probability-weighted moments.  
*Constraint:* OPTOPT = -2, -1, 1 or 2.
- 4: XI – REAL (KIND=nag\_wp) *Output*  
*On exit:* the parameter estimate  $\hat{\xi}$ .
- 5: BETA – REAL (KIND=nag\_wp) *Output*  
*On exit:* the parameter estimate  $\hat{\beta}$ .
- 6: ASVC(4) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the variance-covariance of the asymptotic Normal distribution of  $\hat{\xi}$  and  $\hat{\beta}$ . ASVC(1) contains the variance of  $\hat{\xi}$ ; ASVC(4) contains the variance of  $\hat{\beta}$ ; ASVC(2) and ASVC(3) contain the covariance of  $\hat{\xi}$  and  $\hat{\beta}$ .
- 7: OBSVC(4) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* if maximum likelihood estimates are requested, the observed variance-covariance of  $\hat{\xi}$  and  $\hat{\beta}$ . OBSVC(1) contains the variance of  $\hat{\xi}$ ; OBSVC(4) contains the variance of  $\hat{\beta}$ ; OBSVC(2) and OBSVC(3) contain the covariance of  $\hat{\xi}$  and  $\hat{\beta}$ .

- 8: LL – REAL (KIND=nag\_wp) *Output*  
*On exit:* if maximum likelihood estimates are requested, LL contains the log-likelihood value  $L$  at the end of the optimization; otherwise LL is set to  $-1.0$ .
- 9: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0,  $-1$  or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value  $-1$  or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N = \langle value \rangle$ .  
 Constraint:  $N > 1$ .

IFAIL = 2

On entry,  $Y(\langle value \rangle) = \langle value \rangle$ .  
 Constraint:  $Y(i) \geq 0.0$  for all  $i$ .

IFAIL = 3

On entry, OPTOPT =  $\langle value \rangle$ .  
 Constraint: OPTOPT =  $-2, -1, 1$  or  $2$ .

IFAIL = 6

The asymptotic distribution is not available for the returned parameter estimates.

IFAIL = 7

The distribution of maximum likelihood estimates cannot be calculated for the returned parameter estimates because the Hessian matrix could not be inverted.

IFAIL = 8

The asymptotic distribution of parameter estimates is invalid and the distribution of maximum likelihood estimates cannot be calculated for the returned parameter estimates because the Hessian matrix could not be inverted.

IFAIL = 9

The optimization of log-likelihood failed to converge; no maximum likelihood estimates are returned. Try using the other maximum likelihood option by resetting OPTOPT. If this also fails, moments-based estimates can be returned by an appropriate setting of OPTOPT.

IFAIL = 10

Variance of data in Y is too low for method of moments optimization.

IFAIL = 11

The sum of Y is zero within *machine precision*.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

## 7 Accuracy

Not applicable.

## 8 Parallelism and Performance

G07BFF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The search for maximum likelihood parameter estimates is further restricted by requiring

$$1 + \frac{\hat{\xi}y_i}{\hat{\beta}} > 0,$$

as this avoids the possibility of making the log-likelihood  $L$  arbitrarily high.

## 10 Example

This example calculates parameter estimates for 23 observations assumed to be drawn from a generalized Pareto distribution.

### 10.1 Program Text

```

Program g07bffe
!      G07BFF Example Program Text
!
!      Mark 25 Release. NAG Copyright 2014.
!
!      .. Use Statements ..
!      Use nag_library, Only: g07bff, nag_wp
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..

```

```

      Real (Kind=nag_wp)           :: beta, ll, xi
      Integer                     :: ifail, n, optopt
!      .. Local Arrays ..
      Real (Kind=nag_wp)          :: asvc(4), obsvc(4)
      Real (Kind=nag_wp), Allocatable :: y(:)
!      .. Executable Statements ..
      Write (nout,*) 'G07BFF Example Program Results'
      Write (nout,*)

!      Skip header
      Read (nin,*)

!      Read in problem size and control parameters
      Read (nin,*) n, optopt

      Allocate (y(n))

!      Read in data
      Read (nin,*) y(1:n)

!      Calculate the GPD parameter estimates
      ifail = 1
      Call g07bff(n,y,optopt,xi,beta,asvc,obsvc,ll,ifail)
      If (ifail/=0) Then
         If (ifail/=6 .And. ifail/=7 .And. ifail/=8) Then
            Write (*,99997) '** G07BFF returned with IFAIL = ', ifail
            Go To 100
         End If
      End If

!      Display parameter estimates
      Write (nout,*) 'Parameter estimates'
      Write (nout,Fmt=99998) 'xi           ', xi
      Write (nout,Fmt=99998) 'beta          ', beta
      Write (nout,*)

!      Display parameter distribution
      If (optopt>0) Then
         If (ifail==7 .Or. ifail==8) Then
            Write (nout,Fmt=99999) 'Invalid observed distribution'
         Else
            Write (nout,*) 'Observed distribution'
            Write (nout,Fmt=99998) 'Var(xi)           ', obsvc(1)
            Write (nout,Fmt=99998) 'Var(beta)          ', obsvc(4)
            Write (nout,Fmt=99998) 'Covar(xi,beta)     ', obsvc(2)
            Write (nout,Fmt=99998) 'Final log-likelihood:', ll
         End If
         Write (nout,*)

      Else
         If (ifail==6 .Or. ifail==7) Then
            Write (nout,Fmt=99999) 'Invalid asymptotic distribution'
         Else
            Write (nout,*) 'Asymptotic distribution'
            Write (nout,Fmt=99998) 'Var(xi)           ', asvc(1)
            Write (nout,Fmt=99998) 'Var(beta)          ', asvc(4)
            Write (nout,Fmt=99998) 'Covar(xi,beta)     ', asvc(2)
         End If
      End If

100   Continue

99999 Format (1X,A)
99998 Format (1X,A,1X,E14.6)
99997 Format (1X,A,I0)
      End Program g07bff

```

## 10.2 Program Data

G07BFF Example Program Data

```
23 2
1.5800
0.1390
2.3624
2.9435
0.1363
0.9688
0.6585
2.8011
0.9880
1.7887
0.0630
0.3862
1.5130
0.0669
1.3659
0.4256
0.3485
27.8760
5.2503
1.1028
0.5273
1.3189
0.6490
```

## 10.3 Program Results

G07BFF Example Program Results

Parameter estimates

```
xi          0.540439E+00
beta        0.104055E+01
```

Observed distribution

```
Var(xi)          0.799320E-01
Var(beta)        0.119872E+00
Covar(xi,beta)   -0.455092E-01
Final log-likelihood: -0.363443E+02
```

---